DECUS
PROGRAM LIBRARY

DECUS NO.	FOCAL8-89
TITLE	THE RECURSIVE EVALUATION OF FUNCTIONS
AUTHOR	A. K. Head
COMPANY	C.S.I.R.O. Division of Tribophysics Melbourne, Australia
DATE	Submitted: February 8, 1970
SOURCE LANGUAGE	FOCAL

THE RECURSIVE EVALUATION OF FUNCTIONS

DECUS Program Library Write-up — DECUS No. FOCAL8-89

The evaluation of functions by recursion is a technique which is not often used in computing, but is has proved to be the solution to two different problems in FOCAL. The first was the need for circular functions which were more accurate than the internal functions when using 4-word arithmetic (when the internal functions have only 3-word accuracy). The second was the need for functions which occupy less memory than the corresponding internal functions. A selection of some of the circular and hyperbolic functions which have been used are as follows.

```
C-FOCAL,1969

10.01 C    SIN: DO 10.2      COS: DO 10
10.10 S X=1.5707963268-X
10.20 I (X↑2-.01)10.3;S X=X/3;D 10.2;S X=3*X-4*X↑3;R
10.30 S X=X-X↑3/6+X↑5/120

11.01 C    TAN: DO 11
11.10 I (X↑2-.01)11.2;S X=X/2;D 11;S X=2*X/(1-X↑2+1E-99);R
11.20 S X=X+X↑3/3+X↑5/7.5+X↑7/315

12.01 C    ASIN: DO 12       ACOS: D 12.3
12.10 I (X↑2-.01)12.2;S X=X/(FSQT(1+X)+FSQT(1-X));D 12;S X=2*X;R
12.20 S X=X+X↑3/6+.075*X↑5+X↑7/22.4;R
12.30 D 12;D 10.1

13.01 C    ATAN: DO 13
13.10 I (X↑2-.01)13.2;S X=X/(1+FSQT(X↑2+1));D 13;S X=2*X;R
13.20 S X=X-X↑3/3+X↑5/5-X↑7/7

14.01 C    EXP: DO 14
14.10 I (X↑2-.01)14.2;S X=X/2;D 14;S X=X↑2;R
14.20 S X=1+X+X↑2/2+X↑3/6+X↑4/24+X↑5/120+X↑6/720

15.01 C    LOG: DO 15
15.10 I (X↑2-2.04*X+1)15.2;S X=FSQT(X);D 15;S X=2*X;R
15.20 S X=(X-1)/(X+1);S X=2*(X+X↑3/3+X↑5/5+X↑7/7)

16.01 C    SINH: DO 16       COSH: DO 16.3
16.10 I (X↑2-.01)16.2;S X=X/3;D 16;S X=3*X+4*X↑3;R
16.20 S X=X+X↑3/6+X↑5/120;R
16.30 D 16;S X=FSQT(1+X↑2)
```

Angles are in radians, ASIN and ATAN are in the range $-\pi/2$ to $+\pi/2$, ACOS in the range 0 to π.

These are subroutines rather than functions so for example to compute the cosine of an angle the sequence would be

SET X = ANGLE;DO 10

after which X contains the value of the cosine.

All these subroutines work on the same general principle. If the value of X is sufficiently small, then the function is evaluated by a power series and the task is finished. Otherwise X is repeatedly reduced until it is small enough, the power series is evaluated, then the function is built up the same number of times that X was reduced. For instance, EXP is based on the relation

$$\exp(2x) = (\exp(x))^2$$

so that x is repeatedly halved, then the sum of the power series is squared the same number of times; SIN is based on

$$\sin 3x = 3\sin x - 4\sin^3 x$$

and ASIN on

$$\text{asin}(x) = 2\text{asin}\ (x/(\sqrt{(1 + x)} + \sqrt{(1 - x)})).$$

Instead of using a FOCAL variable to count the numbers of reductions of x, the recursive ability of a FOCAL subroutine to DO itself is used for the same result. Since it may not be obvious what happens in such a case, the following TRACE print out of the evaluation of EXP(1) may be illuminating, and to give a clear print out, T! has been inserted at the beginning of line 14.1

```
*? S X=1;D 14;T X
 S X=1;D 14;T !
;I (X↑2-.01)14.2;S X=X/2;D 14;T !
;I (X↑2-.01)14.2;S X=X/2;D 14;T !
;I (X↑2-.01)14.2;S X=X/2;D 14;T !
;I (X↑2-.01)14.2;S X=X/2;D 14;T !
;I (X↑2-.01)14.2;S X=1+X+X↑2/2+X↑3/6+X↑4/24+X↑5/120+X↑6/720
S X=X↑2;R
S X=X↑2;R
S X=X↑2;R
S X=X↑2;R
T X
 0.271827E+01*
```

It is difficult to make a simple statement about the accuracy of these recursive subroutines but it can be expected that there will be errors in the sixth figure when using 3-word FOCAL and in the ninth figure for 4-word FOCAL. Each subroutine (with the COMMENT removed) occupies less than half the memory space of the corresponding FOCAL internal function and with the freedom to have just those functions which a particular program needs. Similar recursive subroutines can be written for all the circular and hyperbolic functions and also for the elliptic functions when there will be a recursive reduction of both the argument and modulus.